
Ziele

• Komponentenbegriff und Klassensignaturen kennen

• Invarianten für Klassen und Komponenten formulieren können

• Operationsspezifikationen (in der Form von Vor- und Nachbedingungen)
schreiben können

• Das Vertragskonzept hinter Operationsspezifikationen verstehen

• Komponentenspezifikationen erstellen können

• Implizite Invarianten aus Klassendiagrammen herleiten können

• Die formale Repräsentation einer Komponentenspezifikation kennen

3.1 Komponenten
Eine Komponente ist ein Subsystem, das durch ein Klassendiagramm beschrieben
wird. (Wir betrachten nur die interne Sicht von Komponenten mit indirekter
Implementierung.)

<<component>>
M

B C

A D

Annahmen
• Eine Klasse enthält keine zwei Operationen mit demselben Namen und

derselben Anzahl von Parametern.
• Beim Überschreiben (Redefinieren) von Operationen in Subklassen wird die

Sichtbarkeit erhalten.

Komponente für Punkte und Formen

<<component>>
PointsAndShapes

+createCPoint(x:Real, y:Real, c:Real):CPoint
+createShape():Shape
+addPointToShape(p:Point, s:Shape)
+removePointFromShape(p:Point, s:Shape)

+moveShape(s:Shape, mx:Real, my:Real)
+movePoint(p:Point, mx:Real, my:Real)

#xx:Real
#yy:Real

~Point()
~Point(x:Real, y:Real)
~setOwner(s:Shape)
~unSetOwner()
~move(mx:Real, my:Real)

Shape

~Shape()
~addPoint(p:Point)
~removePoint(p:Point)

~move(mx:Real, my:Real)
~belongsTo(p:Point):Boolean {query}

CPoint

~CPoint()
~CPoint(x:Real, y:Real, c:Real)
~setColour(c:Real)

Point

−colour:Real

1

* −pointSet
*

−points #owner

0..1

System

+createPoint(x:Real, y:Real):Point
+System()

* −figures

1

3.2 Klassensignaturen

Definition (Klassensignatur):
Sei ∆ ein Klassendiagramm. Die Klassensignatur von ∆ ist definiert durch
Σ∆ = (S∆,≤, OP∆,Visibility) wobei
1. S∆ = SOCL

∆ ∪ {Void}
2. ≤ ist die partielle Ordnung auf SOCL

∆ (vgl. Kapitel 2)
3. OP∆ = OPOCL

∆ ∪ M∆ ∪ Q∆ ∪ Con∆ wobei

M∆ enthält ◦ für jede Methode m(x1 : T1, . . . , xn : Tn) einer Klasse C
m : C × T1 × . . .× Tn → Void

◦ für jede Methode m(x1 : T1, . . . , xn : Tn) : T einer Klasse C
m : C × T1 × . . .× Tn → T

Q∆ enthält ◦ für jede Query q(x1 : T1, . . . , xn : Tn) : T {query} einer Klasse C
q : C × T1 × . . .× Tn → T

Con∆ enthält ◦ für jeden Konstruktor C(x1 : T1, . . . , xn : Tn) einer Klasse C
C : T1 × . . .× Tn → C

4. Visibility : A∆ ∪M∆ ∪Q∆ ∪ Con∆ → {+,∼,#,−} ist eine Funktion wobei

+ steht für komponenten-öffentliche Sichtbarkeit
∼ steht für komponenten-private Sichtbarkeit
# steht für geschützte Sichtbarkeit
− steht für klassen-private Sichtbarkeit

Notation
Opns∆ =def M∆ ∪Q∆ ∪ Con∆ bezeichnet die Operationen aus ∆

Beispiel (Punkte und Formen):

V isibility(op) = + für alle Operationen op der Klasse System
V isibility(op) = ∼ für alle übrigen Operationen
V isibility( .xx : Point → Real) = #
V isibility( .yy : Point → Real) = #
V isibility( .owner : Point → Shape) = #
V isibility(a) = − für alle übrigen Attribute und Rollennamen

M∆ = {setColour : CPoint × Real → Void
setOwner : Point × Shape → Void ,
unSetOwner : Point → Void ,
move : Point × Real × Real → Void ,
addPoint : Shape × Point → Void ,
removePoint : Shape × Point → Void ,
move : Shape × Real × Real → Void ,
createPoint : System × Real × Real → Point ,
. . .}

Q∆ = {belongsTo : Shape × Point → Boolean}
Con∆ = {CPoint : → CPoint

CPoint : Real × Real × Real → CPoint ,
Point : → Point ,
Point : Real × Real → Point ,
Shape : → Shape,
System : → System}

3.3 Klassen- und Komponenteninvarianten
Sei ∆ ein Klassendiagramm.

• Für jede Klasse in ∆ kann eine Klasseninvariante angegeben werden.
• Eine Klasseninvariante ist ein OCL-Ausdruck vom Typ Boolean, der die

möglichen Zustände, unter denen Objekte der betreffenden Klasse von
anderen Objekten aus gesehen werden können, einschränkt.

Beispiel (Zähler):

Counter MCounter

count: Integer last: Integer

... ...

(InvCounter) context Counter
inv : count >= 0

(InvMCounter) context MCounter
inv : last >= 0

c0:Counter m0:MCounter

count = 7 count = 0
last = 3

Ein Zustand, in dem die
Klasseninvarianten erfüllt sind.

Allgemeine Form von Klasseninvarianten
Eine Invariante für eine Klasse C in ∆ wird spezifiziert durch

context C
inv : Inv

oder
durch

context C
inv invname : Inv

wobei Inv ∈ EXPOCL
Boolean mit FV (Inv) ⊆ {self }.

Komponenteninvarianten

<<component>>
M

B C

A D

• Für jede Komponente M mit Klassendiagramm ∆ kann eine
Komponenteninvariante angegeben werden.

• Eine Komponenteninvariante ist ein OCL-Ausdruck vom Typ Boolean, der die
möglichen Objektkonfigurationen einschränkt, die von außerhalb der
Komponente gesehen werden können.

Beispiel (Punkte und Formen):

Die folgende Komponenteninvariante verlangt, dass assoziierte Punkte und
Formen zum selben System gehören:

context PointsAndShapes
inv invSameSystem:

System.allInstances() -> forAll(sys |
sys.pointSet -> forAll(p |
p.owner <> null implies
sys.figures -> includes(p.owner)) and

sys.figures -> forAll(s |
s.points -> forAll(p |
sys.pointSet -> includes(p))))

Beispiel (Seminar):

<<component>>
Seminar

Instructor

Course

Session

qualifiedFor*

*

1

*

sessions

1

~changeCourse(c: Course)

*

for

...

(InvSeminar) context Seminar
inv : Instructor.allInstances()->forAll(i|

i.qualifiedFor->
includesAll(i.sessions.for->asSet()))

Beispiel (Seminar):
Ein Zustand, in dem die Komponenteninvariante erfüllt ist:

i0: Instructor

c2: Course

c1: Coursec0: Course

qualifiedForqualifiedFor s0: Session

sessions

s1: Session

sessions

for

for

Allgemeine Form von Komponenteninvarianten
Eine Komponenteninvariante für eine Komponente M mit Klassendiagramm ∆ wird
spezifiziert durch

context M
inv : Inv

oder
durch

context M
inv invname : Inv

wobei Inv ∈ EXPOCL
Boolean mit FV (Inv) = ∅.

3.4 Operationsspezifikationen
Sei ∆ ein Klassendiagramm.

• Für jede Operation op ∈ Opns∆ kann eine Operationsspezifikation angegeben
werden, welche das mögliche Verhalten der Operation einschränkt.

• Eine Operationsspezifikation ist in der Form von Vor- und Nachbedingungen
gegeben:

◦ Die Vorbedingung muss wahr sein, wenn die Operation aufgerufen wird.
◦ Die Nachbedingung muss wahr sein, wenn die Operation beendet ist (i.e. nach Ausführung

der Operation).

Beispiel (Punkte und Formen):

context System::addPointToShape(p:Point, s:Shape)
pre: p<>null and s<>null and

pointSet -> includes(p) and
figures -> includes(s) and
p.owner = null and s.points -> excludes(p)

post: p.owner = s and
s.points = s.points@pre -> including(p)

context System::createPoint(x:Real, y:Real):Point
post: result.oclIsNew() and

result.xx = x and result.yy = y and
result.owner = null and
pointSet = pointSet@pre -> including(result)

context Point::move(mx:Real, my:Real)
post: xx = xx@pre + mx and

yy = yy@pre + my

context Shape::belongsTo(p:Point):Boolean
post: result = points -> includes(p)

context CPoint::CPoint(x:Real, y:Real, c:Real)
post: xx = x and yy = y and colour = c

context Point::setOwner(s:Shape)
pre: s<>null and owner = null

post: owner = s

context Shape::addPoint(p:Point)
pre: p<>null and points -> excludes(p)

post: points = points@pre -> including(p)

Allgemeine Form von Operationsspezifikationen

• Für Methoden (m : C × T1 × . . .× Tn → Void) ∈ M∆ ohne Rückgebewert:

context C :: m(x1 : T1, . . . , xn : Tn)
pre : P
post : Q

so dass
◦ P ∈ EXPOCL

Boolean, Q ∈ EEXPOCL
Boolean

◦ FV (P) ⊆ {self : C , x1 : T1 , . . . , xn : Tn}
◦ FV (Q) ⊆ {self : C , x1 : T1 , . . . , xn : Tn}

• Für Methoden (m : C × T1 × . . .× Tn → T ) ∈ M∆ mit Rückgabewert:

context C :: m(x1 : T1, . . . , xn : Tn) : T
pre : P
post : Q

so dass
◦ P ∈ EXPOCL

Boolean, Q ∈ EEXPOCL
Boolean

◦ FV (P) ⊆ {self : C , x1 : T1 , . . . , xn : Tn}
◦ FV (Q) ⊆ {self : C , x1 : T1 , . . . , xn : Tn, result : T}

• Für Queries (q : C × T1 × . . .× Tn → T ) ∈ Q∆ haben Operations-
spezifikationen dieselbe Form wie für Methoden mit Rückgabewert.

• Für Konstruktoren (C : T1 × . . .× Tn → C ) ∈ Con∆:

context C :: C(x1 : T1, . . . , xn : Tn)
pre : P
post : Q

so dass
◦ P ∈ EXPOCL

Boolean, Q ∈ EEXPOCL
Boolean

◦ FV (P) ⊆ {x1 , . . . , xn}
◦ FV (Q) ⊆ {self : C , x1 : T1 , . . . , xn : Tn}

Anmerkung:

Vor- und Nachbedingungen kann ein benutzerdefinierter Name gegeben werden:

context C :: op(x1 : T1, . . . , xn : Tn) : T
pre prename : P
post postname : Q

Frame Assumption

Unter einer “Frame Assumption” versteht man eine implizite Verstärkung der
Nachbedingung, die besagt, dass außer den explizit geforderten
Zustandsänderungen keine weiteren Änderungen des Vorzustands erfolgen
dürfen.

Operationsspezifikationen als Verträge

Eine Operationsspezifikation kann als ein Vertrag zwischen
• einem Kunden, der die Operation benutzt, und
• einem Programmierer, der die Operation implementiert,
betrachtet werden.

Beide einigen sich, dass sie die folgenden Verpflichtungen erfüllen:

Verpflichtung des Kunden:
Der Kunde ruft die Operation nur auf, wenn die Vorbedingung erfüllt ist.

Verpflichtung des Programmierers:
Unter der Voraussetzung, dass die Operation in einem Zustand aufgerufen wird, in
dem die Vorbedingung erfüllt ist, sichert der Programmierer zu:
• Die Operation terminiert und verursacht keinen Laufzeitfehler.
• Nach Ausführung der Operation gilt die Nachbedingung.

Beachte:

1. Der Vertrag sagt nichts über Situationen aus, in denen die Vorbedingung nicht
erfüllt ist.

2. Falls die Operation eine Query ist, dann sichert der Programmierer zusätzlich
zu, dass sich der Zustand des Systems während der Ausführung der Operation
nicht ändert.

3. Falls die Operation ein Konstruktor ist, dann sichert der Programmierer
zusätzlich zu, dass nach Ausführung des Konstruktors ein neues Objekt erzeugt
worden ist.

Bemerkung:

• Je stärker die Vorbedingung ist, desto einfacher wird die “Arbeit” für den
Implementierer und desto schwieriger wird die “Arbeit” für den Benutzer.

• Je stärker die Nachbedingung ist, desto schwieriger wird die “Arbeit” für den
Implementierer und desto einfacher wird die “Arbeit” für den Benutzer.

Vertragszusätze im Kontext von Invarianten

Für komponenten-private Operationen:
Für den Kunden:
Der Kunde verpflichtet sich zusätzlich die Operation nur dann aufzurufen, wenn alle
Klasseninvarianten erfüllt sind.

Für den Programmierer:
Der Programmierer kann auch vorraussetzen, dass beim Operationsaufruf alle
Klasseninvarianten erfüllt sind, verpflichtet sich aber zusätzlich dafür zu sorgen, dass nach
Ausführung der Operation wieder alle Klasseninvarianten erfüllt sind.

Für komponenten-öffentliche Operationen:
Für den Kunden:
Der Kunde verpflichtet sich zusätzlich die Operation nur dann aufzurufen, wenn alle Klassen-
und Komponenteninvarianten erfüllt sind.

Für den Programmierer:
Der Programmierer kann auch vorraussetzen, dass beim Operationsaufruf alle Klassen- und
Komponenteninvarianten erfüllt sind, verpflichtet sich aber zusätzlich dafür zu sorgen, dass
nach Ausführung der Operation wieder alle Klassen- und Komponenteninvarianten erfüllt
sind.

Bemerkung:

Unter bestimmten Voraussetzungen, die später untersucht werden (keine
öffentlichen Attribute, Lokalitätsprinzip von Klasseninvarianten), können die
Verpflichtungen beider Beteiligter zur Berücksichtigung der Invarianten vereinfacht
werden.

Methodisches Vorgehen bei der Erstellung von Operationsspezifikationen

1. Die offensichtlichen Vorbedingungen formulieren.
2. Die gewünschten Nachbedingungen angeben.
3. Die Vorbedingungen ggf. verstärken um die Verträglichkeit mit den Invarianten

zu gewährleisten.

3.5 Komponentenspezifikationen
Eine Komponentenspezifikation

CompSpec = (〈M ,∆〉, Invse,OpSpecs)

besteht aus

• einer Komponente M mit Klassendiagramm ∆,
• einer Menge Invse von (expliziten) Klasseninvarianten und (expliziten)

Komponenteninvarianten und
• einer Menge OpSpecs von Operationsspezifikationen

für die Operationen aus Opns∆.

Annahme:
OpSpecs enthält für jede Operation op ∈ Opns∆ genau eine Operationsspezifikation.
(Default: context C::op(x1 : T1,...,xn : Tn) pre: true post: true)

Komposition von Operationsspezifikationen

Gegeben seien zwei Operationsspezifikationen für dieselbe Operation:

(Spec1) context C :: op(x1 : T1, . . . , xn : Tn) : T
pre : P1

post : Q1

(Spec2) context C :: op(x1 : T1, . . . , xn : Tn) : T
pre : P2

post : Q2

1. Join-Operator

Join(Spec1, Spec2) ist definiert durch

context C :: op(x1 : T1, . . . , xn : Tn) : T

pre : P1 and P2

post : Q1 and Q2

Beispiel (Join) Counter1

count: Integer
last: Ingeger

dec()
...

(Spec1) context Counter1::dec()
pre: count > 0
post: count = count@pre - 1

(Spec2) context Counter1::dec()
pre: true
post: last = count@pre

Join(Spec1,Spec2) context Counter1::dec()
pre: count > 0
post: count = count@pre - 1 and

last = count@pre

2. Combine-Operator

Combine(Spec1, Spec2) ist definiert durch

context C :: op(x1 : T1, . . . , xn : Tn) : T

pre : P1 or P2

post : (P1@pre implies Q1) and
(P2@pre implies Q2)

wobei Pi@pre (für i = 1, 2) den OCL-Ausdruck bezeichnet, der aus Pi entsteht,
wenn man
• alle Vorkommen von Namen a mit ( .a : C → T ) ∈ A∆ ersetzt durch a@pre und
• alle Vorkommen von Ausdrücken D.allInstances() ersetzt durch

D.allInstances@pre().

Beispiel (Combine)

count: Integer

Counter2

...
dec()

(Spec1) context Counter2::dec()
pre: count > 0

post: count = count@pre - 1

(Spec2) context Counter2::dec()
pre: count = 0

post: count = 0

Combine(Spec1,Spec2)

context Counter2::dec()
pre: count >= 0

post: (count@pre > 0 implies count = count@pre - 1) and
(count@pre = 0 implies count = 0)

Komponentenspezifikation für Punkte und Formen

<<component>>
PointsAndShapes

+createCPoint(x:Real, y:Real, c:Real):CPoint
+createShape():Shape
+addPointToShape(p:Point, s:Shape)
+removePointFromShape(p:Point, s:Shape)

+moveShape(s:Shape, mx:Real, my:Real)
+movePoint(p:Point, mx:Real, my:Real)

#xx:Real
#yy:Real

~Point()
~Point(x:Real, y:Real)
~setOwner(s:Shape)
~unSetOwner()
~move(mx:Real, my:Real)

Shape

~Shape()
~addPoint(p:Point)
~removePoint(p:Point)

~move(mx:Real, my:Real)
~belongsTo(p:Point):Boolean {query}

CPoint

~CPoint()
~CPoint(x:Real, y:Real, c:Real)
~setColour(c:Real)

Point

−colour:Real

1

* −pointSet
*

−points #owner

0..1

System

+createPoint(x:Real, y:Real):Point
+System()

* −figures

1

• Es wird keine explizite (nicht-triviale) Klasseninvariante gefordert.
• Es wird die explizite Komponenteninvariante invSameSystem von Abschnitt 3.3

verwendet, die verlangt, dass assoziierte Punkte und Formen zum selben
System gehören:

context PointsAndShapes
inv invSameSystem:

System.allInstances() -> forAll(sys |
sys.pointSet -> forAll(p |

p.owner <> null implies
sys.figures -> includes(p.owner)) and

sys.figures -> forAll(s |
s.points -> forAll(p |
sys.pointSet -> includes(p))))

• Als Operationsspezifikationen werden die in Abschnitt 3.4 angegebenen Vor-
und Nachbedingungen verwendet, ergänzt um geeignete Spezifikationen für die
restlichen Operationen.

3.6 Formale Repräsentation von Komponentenspezifikationen

Motivation

• Ein Klassendiagramm ∆ drückt Bedingungen für Assoziationen durch
Multiplizitäten und Bidirektionalitäts-Anforderungen aus.

• Solche Bedingungen führen zu ”impliziten” Klassen- und
Komponenteninvarianten, die berücksichtigt werden müssen, wenn man eine
präzise Semantik von Komponentenspezifikationen angeben sowie über die
Korrektheit von Realisierungen (d.h. Programmen) sprechen will.

Definition:
Sei CompSpec = (〈M ,∆〉, Invse,OpSpecs) eine Komponentenspezifikation.
Die formale Repräsentation von CompSpec ist gegeben durch

FRep(CompSpec) = (〈M ,Σ∆〉, Invs,OpSpecs)

wobei

• Σ∆ die Klassensignatur von ∆ ist und
• Invs = Invse ∪ Invs i die Menge aller expliziten und impliziten Invarianten ist.

Im Folgenden wird beschrieben, wie implizite Klassen- und
Komponenteninvarianten hergeleitet werden.

Anmerkung:

〈M,∆〉 enthält dieselbe Information wie 〈M,Σ∆〉 zusammen mit Invs i .

Implizite Klassen- und Komponenteninvarianten
Multiplizitäten und bidirektionale Assoziationen induzieren Bedingungen für
Objektzustände und Objektkonfigurationen, die durch implizite Klassen- und
Komponenteninvarianten formalisiert werden.

Multiplizitäten an den Enden unidirektionaler Assoziationen

A BmultA
myB

multB

<<subsystem>>
M

R. Hennicker: Formale Objektorientierte Software-Entwicklung 01.06.2006

1. Klasseninvarianten, die durch Multiplizitäten an
navigierbaren Assoziationsenden induziert werden

Fall 1.1: multB = 0 ..1
ist bereits vollständig formalisiert durch das Symbol .myB : A → B

Fall 1.2: multB = 1
induziert die Klasseninvariante

context A
inv : myB <> null

Fall 1.3: multB = ∗
ist bereits vollständig formalisiert durch das Symbol .myB : A → Set(B)

Fall 1.4: multB = 1 ..∗
induziert die Klasseninvariante

context A
inv : myB → exists(b | b <> null)

2. Komponenteninvarianten, die durch durch Multiplizitäten an
nicht navigierbaren Assoziationsenden induziert werden

Fall 2.1: multA = 0 ..1 , multB ≤ 1 induziert die Komponenteninvariante
context M

inv : B .allInstances() → forAll(b |
A.allInstances() → select(a |

a.myB = b) → size() <= 1 )

Fall 2.2: multA = 1 , multB ≤ 1 induziert
context M

inv : B .allInstances() → forAll(b |
A.allInstances() → one(a | a.myB = b))

Fall 2.3: multA = ∗ induziert keine Einschränkung

Fall 2.4: multA = 1 ..∗, multB ≤ 1 induziert
context M

inv : B .allInstances() → forAll(b |
A.allInstances() → exists(a | a.myB = b))

Fall 2.5: Falls multB > 1 dann ersetze in den obigen Fällen
a.myB = b durch a.myB → includes(b)

Beispiel (Punkte und Formen):

SystemPoint Shape
pointSet
* 1 1 *

figures

induziert die folgende Komponenteninvariante:

context PointsAndShapes
inv invOneSystem:

Point.allInstances() -> forAll(p |
System.allInstances() -> one(sys |
sys.pointSet -> includes(p))) and

Shape.allInstances() -> forAll(s |
System.allInstances() -> one(sys |
sys.figures -> includes(s)))

Bidirektionale Assoziationen

M
<<subsystem>>

BA
myA myB

multBmultA

induzieren die folgenden impliziten Invarianten:

1. Klasseninvariante für A induziert durch die Multiplizität am navigierbaren
Assoziationsende mit Rollennamen myB (wie im unidirektionalen Fall).

2. Klasseninvariante für B induziert durch durch die Multiplizität am navigierbaren
Assoziationsende mit Rollennamen myA (wie im unidirektionalen Fall).

3. Komponenteninvariante für M induziert durch die Anforderung
der Bidirektionalität.

Fall 3.1: multA = multB = 0 ..1
induziert die Komponenteninvariante

context M
inv : A.allInstances() → forAll(a |

a.myB <> null implies a.myB .myA = a) and
B .allInstances() → forAll(b |

b.myA <> null implies b.myA.myB = b)

Fall 3.2: multA = multB = 1
induziert die Komponenteninvariante

context M
inv : A.allInstances() → forAll(a | a.myB .myA = a) and

B .allInstances() → forAll(b | b.myA.myB = b)

Fall 3.3: multA = 0 ..1 , multB = 1 oder multA = 1 , multB = 0 ..1
ist eine Mischung aus den Fällen 3.1 und 3.2

Fall 3.4: multA = ∗ und multB = ∗
induziert die Komponenteninvariante

context M
inv : A.allInstances() → forAll(a |

a.myB → forAll(b | b <> null implies b.myA → includes(a)) and
B .allInstances() → forAll(b |

b.myA → forAll(a | a <> null implies a.myB → includes(b))

Alle übrigen Fälle sind geeignete Mischformen der oben betrachteten Fälle.

Beispiel (Punkte und Formen):

points owner
0..1Point * Shape

induziert die folgende Komponenteninvariante:

context PointsAndShapes
inv invBidirect:

Point.allInstances() -> forAll(p |
p.owner <> null implies
p.owner.points -> includes(p)) and

Shape.allInstances() -> forAll(s |
s.points -> forAll(p |
p <> null implies p.owner = s))

Formale Repräsentation der Komponente für Punkte und Formen

(〈PointsAndShapes,Σ∆〉, Invs,OpSpecs)

wobei

• Σ∆ die Klassensignatur aus Abschnitt 3.2 ist,

• Invs = {invSameSystem, invOneSystem, invBidirect},

• OpSpecs die in Abschnitt 3.4 angegebenen Operationsspezifikationen sind.

3.7 Zusammenfassung

• Jedem Klassendiagramm ∆ kann eine Klassensignatur
Σ∆ = (S∆,≤, OP∆,Visibility) zugeordnet werden, die eine Erweiterung der
OCL-Signatur ΣOCL

∆ ist.

• Eine Klasseninvariante spezifiziert eine Eigenschaft, die von allen Objekten der
Klasse ”von außen” betrachtet, erfüllt sein muss.

• Eine Komponenteninvariante spezifiziert eine Eigenschaft, die von allen
Objektkonfigurationen erfüllt sein muss, wenn man die Komponente
”von außen” betrachtet.

• Für alle Operationen op ∈ Opns∆ = M∆ ∪Q∆ ∪ Con∆ können
Operationsspezifikationen in der Form von Vor- und Nachbedingungen
angegeben werden.

• Eine Operationsspezifikation kann als Vertrag zwischen dem Benutzer der
Operation (Kunde) und dem Programmierer der Operation angesehen werden.

• Eine Komponentenspezifikation CompSpec = (〈M ,∆〉, Invse,OpSpecs)
besteht aus einer Komponente mit Klassendiagramm, einer Menge von
expliziten Klassen- und Komponenteninvarianten und einer Menge von
Operationsspezifikationen.

• Jede Komponentenspezifikation hat eine formale Repräsentation
FRep(CompSpec) = (〈M ,Σ∆〉, Invs,OpSpecs) wobei Invs die expliziten
Invarianten um implizite Invarianten ergänzt, die sich aus den Multiplizitäten und
bidirektionalen Assoziationen von ∆ ergeben.

: June 20, 2006

